Tutorial 5: Linear Algebra in *Mathematica* Part 1

```
Off[General::spell];
```

Version 1, BRW, 4/23/07

I wish to perform a few simple matrix mechanics using *Mathematica*. Lets define a 3x3 matrix A, B, and vector c.

```
A = {{4, 0, 1}, {-2, 1, 0}, {-2, 0, 1}};
B = {{0, 1, 1}, {-3, 1, 5}, {0, 0, 1}};
c = {1, 2, 3};
```

We can look at it in Matrix Form

```
MatrixForm[A]
MatrixForm[B]
MatrixForm[c]
```

$$\begin{pmatrix} 4 & 0 & 1 \\ -2 & 1 & 0 \\ -2 & 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 0 & 1 & 1 \\ -3 & 1 & 5 \\ 0 & 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 \\ 2 \\ 3 \end{pmatrix}$$

We can get the individual elements of the matrix using the [[ ]] notation, where A[[i,j]] is the element A_{ij}. Notice that c only has one index.

```
A[[1, 1]]
```

4

```
B[[3, 2]]
```

0

```
c[[3]]
```

3

We can also define an nxn identity matrix.

```
ii = IdentityMatrix[3];
MatrixForm[ii]
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

We can transpose the matrix

```
Transpose[A] // MatrixForm
```

$$\begin{pmatrix} 4 & -2 & -2 \\ 0 & 1 & 0 \\ 1 & 0 & 1 \end{pmatrix}$$

■ Matrix Multiplication

We can easily multiply matrices using the "." operator. Notice that A.B is another 3x3 matrix and B.c is a 3x1 vector.

```
A.B
```

```
{{0, 4, 5}, {-3, -1, 3}, {0, -2, -1}}
```

```
MatrixForm[A.B]
```

$$\begin{pmatrix} 0 & 4 & 5 \\ -3 & -1 & 3 \\ 0 & -2 & -1 \end{pmatrix}$$

```
MatrixForm[B.c]
```

$$\begin{pmatrix} 5 \\ 14 \\ 3 \end{pmatrix}$$

■ Eigenvalues and Eigenvectors

To find the eigenvalues and eigenvectors of A we use the command Eigenvalues and Eigenvectors. We define λ1, λ2, and λ3 as the eigenvalues.

```
eval = Eigenvalues[A]
λ1 = eval[[1]]
λ2 = eval[[2]]
λ3 = eval[[3]]
```

```
{3, 2, 1}

3

2

1
```

For the eigenvectors we define x1 x2 and x3

```
evec = Eigenvectors[A]
x1 = evec[[1]]
x2 = evec[[2]]
x3 = evec[[3]]
```

```
{{-1, 1, 1}, {-1, 2, 2}, {0, 1, 0}}

{-1, 1, 1}

{-1, 2, 2}

{0, 1, 0}
```

Now verify the eigenvectors using Ax=λx, we get all true statements.

```
A.x1 == λ1 x1
A.x2 == λ2 x2
A.x3 == λ3 x3
```

```
True

True

True
```

We can also solve for the eigenvalues by computing the determinant of A-λ I

```
A - λ ii // MatrixForm
```

$$\begin{pmatrix} 4-\lambda & 0 & 1 \\ -2 & 1-\lambda & 0 \\ -2 & 0 & 1-\lambda \end{pmatrix}$$

```
Det[A - λ ii] == 0
```

$6 - 11\lambda + 6\lambda^2 - \lambda^3 == 0$

We can solve this equation using Solve

```
sol = Solve[Det[A - λ ii] == 0, λ]
λ1 = λ /. sol[[1]]
λ2 = λ /. sol[[2]]
λ3 = λ /. sol[[3]]
```

```
{{λ → 1}, {λ → 2}, {λ → 3}}

1

2

3
```

■ Solving for a system of n equations and n unknowns

I wish to solve 3 equations for three unknowns for x, y, z given.

x+2 z=6
-3 x+4 y+6 z=30
-x-2 y+3 z=8

I will solve these using the Solve function and the Cramers rule. First I will use solve and matrix notation to write the equations in a matrix form

■ Using Solve

```
A = {{1, 0, 2}, {-3, 4, 6}, {-1, -2, 3}};
v = {x, y, z};
c = {6, 30, 8};
```

```
A // MatrixForm
eqs = A.v;
```

$$\begin{pmatrix} 1 & 0 & 2 \\ -3 & 4 & 6 \\ -1 & -2 & 3 \end{pmatrix}$$

Write out the equations I wish to solve

```
eqs[[1]] == c[[1]]
eqs[[2]] == c[[2]]
eqs[[3]] == c[[3]]
```

$x + 2 z == 6$

$-3 x + 4 y + 6 z == 30$

$-x - 2 y + 3 z == 8$

Use Solve to find x, y, and z

```
Solve[{eqs[[1]] == c[[1]], eqs[[2]] == c[[2]], eqs[[3]] == c[[3]]}, {x, y, z}]
```

$$\left\{\left\{x \to -\frac{10}{11}, y \to \frac{18}{11}, z \to \frac{38}{11}\right\}\right\}$$

■ Using the Cramer's Rule

These are our solution for x, y, and z. Lets use the Cramer's rule to solve instead. All we need to do is find two determinants. The second matrix A1 is just A with its 1st column replaced with the vector c. The matrix A2 is just A with its 2st column replaced with the vector c. Same for A3. To find the solution x_i we just need to find the ratio

$$x_i = \frac{\text{Det[A1]}}{\text{Det[A]}}$$

Lets do this for the system of equations
x+2 z==6
-3 x+4 y+6 z==30
-x-2 y+3 z==8

```
A1 = {{6, 0, 2}, {30, 4, 6}, {8, -2, 3}};
A2 = {{1, 6, 2}, {-3, 30, 6}, {-1, 8, 3}};
A3 = {{1, 0, 6}, {-3, 4, 30}, {-1, -2, 8}};
MatrixForm[A1]
MatrixForm[A2]
MatrixForm[A3]
```

$$\begin{pmatrix} 6 & 0 & 2 \\ 30 & 4 & 6 \\ 8 & -2 & 3 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 6 & 2 \\ -3 & 30 & 6 \\ -1 & 8 & 3 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 6 \\ -3 & 4 & 30 \\ -1 & -2 & 8 \end{pmatrix}$$

Find the solutions, which are the same we got from Solve. The solve for the others

$$\frac{\text{Det[A1]}}{\text{Det[A]}}$$
$$\frac{\text{Det[A2]}}{\text{Det[A]}}$$
$$\frac{\text{Det[A3]}}{\text{Det[A]}}$$

$$-\frac{10}{11}$$

$$\frac{18}{11}$$

$$\frac{38}{11}$$